```
> with(LinearAlgebra) :
> # funzione ausiliaria:
  # costruisce lista 1,2,...,n escludendo i
  # utile per la costruzione dei cofattori
  #
  complemento := proc(i,n)
    local j ;
    return [seq(j,j=1..i-1),seq(j,j=i+1..n)] ;
  end proc ;
```

complemento := **proc**(i, n) **(1)**

local j; **return** $[seq(j, j = 1 .. i-1), seq(j, j = i+1 .. n)]$

end proc

```
> #
  # costruisce la matrice minore eliminando la riga i
  # e la colonna j dalla matrice A
  #
  minore := proc(A,i,j)
    local r, c, N ;
    # calcola la dimensione della matrice
    # (la matrice e' assunta quadrata)
    N := RowDimension(A) ;
    # costruisce la lista dei numeri di riga del minore
    r := complemento(i,N) ;
    # costruisce la lista dei numeri di colonna del minore
    c := complemento(j,N) ;
    # usa il comando Maple SubMatrix per estrarre il minore
    return SubMatrix(A,r,c) ;
  end proc ;
```

minore := **proc**(A, i, j) **(2)**

local r, c, N;

N := *LinearAlgebra*:-*RowDimension*(A);

r := *complemento*(i, N);

c := *complemento*(j, N);

return *LinearAlgebra*:-*SubMatrix*(A, r, c)

end proc

```
> #
  # calcolo ricorsivo del determinante
  # con lo sviluppo di Laplace
  #
  det := proc(A)
    local j, N, res, sng ;
    # calcola la dimensione della matrice
    N := RowDimension(A) ;
```

```
    #
    # se la matrice e' 1x1 il determinante e' immediato
    if N=1 then return A[1,1] ; end if ;
    #
    # altrimenti uso lo sviluppo di Laplace
    res := 0 ;
    sng := 1 ;
    for j from 1 to N do
      res := res+sng*A[1,j]*det(minore(A,1,j)) ;
      sng := - sng ;
    end do ;
    return res ;
  end proc ;
```

$det := \mathbf{proc}(A)$ **(3)**

$\quad\mathbf{local}\ j, N, res, sng;$

$\quad N := LinearAlgebra\text{:-}RowDimension(A);$

$\quad\mathbf{if}\ N = 1\ \mathbf{then}\ \mathbf{return}\ A[1, 1]\ \mathbf{end\ if};$

$\quad res := 0;$

$\quad sng := 1;$

$\quad\mathbf{for}\ j\ \mathbf{to}\ N\ \mathbf{do}\ res := res + sng * A[1, j] * det(minore(A, 1, j));\ sng := -sng\ \mathbf{end\ do};$

$\quad\mathbf{return}\ res$

$\mathbf{end\ proc}$

```
> # definisce la matrice per gli esempi d'uso
  A := <<1,2,3,4>|<2,2,2,2>|<1,0,1,0>|<2,1,3,5>> ;
```

$$A := \begin{bmatrix} 1 & 2 & 1 & 2 \\ 2 & 2 & 0 & 1 \\ 3 & 2 & 1 & 3 \\ 4 & 2 & 0 & 5 \end{bmatrix} \tag{4}$$

```
> # estrae il minore che corrisponde a (3,2)
  minore(A,3,2) ;
```

$$\begin{bmatrix} 1 & 1 & 2 \\ 2 & 0 & 1 \\ 4 & 0 & 5 \end{bmatrix} \tag{5}$$

```
> # calcola il determinante
  det(A) ;
```

$$-12 \tag{6}$$

```
> # calcola il determinante con le primitive di Maple
  Determinant(A) ;
```

$$-12 \tag{7}$$

```
> # si puo' anche calcolare il determinante
  # con entrate simboliche
  B := <<2,a,b,c>|<1,1,2,0>|<-1,-a,b,2>|<c,b,a,1>> ;
  det(B) ;
```

$$B := \begin{bmatrix} 2 & 1 & -1 & c \\ a & 1 & -a & b \\ b & 2 & b & a \\ c & 0 & 2 & 1 \end{bmatrix} \tag{8}$$

$$11\,b - a\,(b - 2\,a) - a\,(b - a\,c) - b\,(2\,b - b\,c) - 2\,a - a\,c + 2\,b\,c - c\,(4\,a - 2\,b + b\,c + 2\,a\,c)$$

```
> # si puo' anche semplificare
  expand(det(B)) ;
```

$$11\,b - 2\,a\,b + 2\,a^2 + a^2\,c - 2\,b^2 + b^2\,c - 2\,a - 5\,a\,c + 4\,b\,c - b\,c^2 - 2\,a\,c^2 \tag{9}$$

```
> # si puo calcolare il polinomio caratteristico
  poly := det(A+ScalarMatrix(lambda,4,4));
```

$$poly := (1 + \lambda)\,\big((2 + \lambda)\,(1 + \lambda)\,(5 + \lambda) - 2 - 2\,\lambda\big) - 4\,(1 + \lambda)\,(5 + \lambda) + 22 + 20\,\lambda - (2 + \lambda)\,(3 + 3\,\lambda) + 2\,(2 + \lambda)\,(-4 - 4\,\lambda) \tag{10}$$

```
> expand(poly) ;
```

$$-12 - 14\,\lambda + 8\,\lambda^2 + 9\,\lambda^3 + \lambda^4 \tag{11}$$